

ších programových ukázkách. Jiné programovací jazyky ukládají celočíselné hodnoty standardně do 4 bytů, kde 1 bit určuje znaménko a zbývajících 31 bitů obsahuje uloženou hodnotu vyjádřenou ve dvojkové soustavě. Maximální zobrazitelná celočíselná hodnota je v takovém případě rovna $2^{31} - 1 = 2147483647$. Již ve čtvercovém městě velikosti 18×18 ulic bez rozkopaných křižovatek vychází počet možných cest 2333606220 a je tedy vyšší než $2^{31} - 1$. Úloha zadaná v olympiádě vyžadovala napsat program, který zvládne vyřešit všechna města až do velikosti 100×100 ulic. Pro takto velké město bez rozkopaných křižovatek dostaneme výsledek 22750883079422934966181954039568885395604168260154104734000 (což je hodnota řádově 10^{58}). V době konání soutěže před 25 lety ovšem Python ještě neexistoval, takže soutěžící používající programovací jazyky C, C++ nebo Pascal si museli navíc naprogramovat vlastní celočíselnou aritmetiku pracující s mnohacifernými čísly.

Jak moc jsou okružáci okrouhlí?

LUDEK SPÍCHAL

Česká lesnická akademie, Trutnov

Spirály různých tvarů nalezneme ve svém okolí velmi běžně. Přestože spirál je celá řada různých typů, v přírodních útvarech zaznamenáme převážně spirálu logaritmickou.


V minulosti byla u řady zvířat (např. ve tvarech rohů turovitých, špičáků divokých prasat, tvarech ulit a mušlí měkkýšů), v lidské anatomii (např. tvar žeber), ale také v botanice (např. v postavení listů na lodyze, v postavení šupin při tvorbě šišek) prokázána přítomnost logaritmické spirály [4, 5, 8]. Tvar a vlastnosti logaritmické spirály jsou využívány rovněž při konstrukci různých technických zařízení, která vyžadují stálý (tečný) úhel dotyku (např. řezací nástroje, lezecké vačky – tzv. friendly, ozubená kola v kuželových soukolích) [1]. Rozmanitost oblastí výskytu logaritmické spirály je skutečně udivující, např. [2, 3, 8].

Mezi měkkýši vyskytujícími se v České republice nalezneme řadu skupin tvořících spirální ulity s různým způsobem uspořádání závitů. Zástupci čeledi okružákovitých (Planorbidae) vynikají schránkami (obr. 3, 4, 5, 6, 7),

jejichž spirály tvoří závity ležící více méně v jedné rovině. Z tohoto pohledu představují zajímavý materiál pro matematické modelování tvaru schránky.

1. Spirály

Logaritmická spirála je rovinná křivka, jejíž poloměr roste exponenciálně s velikostí úhlu (obr. 1 vlevo, obr. 2). V řadě případů je nejedná pouze o rovinné (2D) křivky, ale křivky v prostoru (3D).¹⁾




Obr. 1 Logaritmická spirála (vlevo), Archimédova spirála (vpravo)

Logaritmickou spirálu můžeme srovnat např. s dobře známou spirálou Archimédovou, která je rovinnou křivkou, jejíž poloměr roste lineárně s velikostí úhlu (obr. 1 vpravo).²⁾

¹⁾Rovinná logaritmická spirála byla objevena nezávisle několika různými matematiky. Jako první se o ní zmiňuje Descartes, který ji v roce 1638 pojmenoval jako ekviangulární (rovnoúhlovou) spirálu, neboť vektor poloměru protíná křivku pod konstantním úhlem (tzv. tečný úhel). Torricelli použil pojmenování geometrická spirála podle další vlastnosti křivky, kdy poloměr křivky roste exponenciálně jako funkce úhlu. Ještě později použil Halley označení proporční spirála, naznačující, že jednotlivé části křivky jsou podobné, liší se proporcí. V roce 1711 Jacob Bernoulli jako první popisuje křivku bez použití polární rovnice, pro její označení použil čtvrtý název – logaritmická spirála.


²⁾Archimédes ze Syrakus (287 př. n. l.–212 př. n. l.) byl řecký matematik, fyzik, filosof, vynálezce a astronom. Archimédova spirála je také označovaná jako aritmetická spirála.



Obr. 2 Dolní špičák (párák) divokého prasete (vlevo), model vnějšího oblouku (vpravo, SAS/STAT) [8]

2. Kdo jsou okružáci?

Čeď okružákovití (Planorbidae) zahrnuje řadu druhů sladkovodních plicnatých plžů, jejichž velikost je v řádech mm nebo jednotek centimetrů.³⁾ Ulity většiny druhů jsou tzv. planispirální, což znamená, že jednotlivé závitky leží zhruba v jedné rovině. Stěny ulit jsou tenké a na povrchu obvykle hladké.



Obr. 3 Mikroskopický snímek plže z čeledi okružákovití (Planorbidae) s vloženou logaritmickou spirálou a vyznačeným tečným úhlem

³⁾ Jistou zvláštností, u plžů nepříliš obvyklou, je přítomnost hemoglobinu (obsahuje železo) jako krevního barviva (většina plžů má v krvi hemocyanin obsahující měď). Přítomnost železa umožňuje efektivnější přenos kyslíku.


V článku se budeme věnovat určení parametrů logaritmických spirál (zejména velikosti tečného úhlu) tvořících schránky okružákovitých pomocí optimalizace softwarem SAS/STAT.

3. Modelování spirál

Pro počítačové modelování je výhodné použití parametrických rovnic zvolených křivek. Parametrická rovnice logaritmické spirály je

$$\begin{aligned} x &= x_0 + r(t) \cos(t), \\ y &= y_0 + r(t) \sin(t), \end{aligned} \tag{1}$$

$t \geq 0$, kde $r(t) = ae^{bt}$ je spojitá funkce (*poloměr spirály*, tj. délka úsečky mezi *pólem spirály* a daným bodem spirály).⁴⁾ Koefficient a určuje bod, od kterého se spirála začíná vykreslovat (*počátek spirály*), pro neposunutou spirálu má souřadnice $[a, 0]$. Pro koefficient b logaritmické spirály platí, že $b = \cot \theta$, kde θ je tečný úhel, který svírá tečna spirály a vektor poloměru v daném bodu.⁵⁾




Obr. 4 Svinutec běloústý (*Anisus leucostoma*). Foto Michal Horsák: <http://mollusca.sav.sk/malacology/img/anisus-leucostoma/>

⁴⁾ Archimédovu spirálu bychom získali pro volbu $r(t) = at$, dále např. pro volbu $r(t) = a\sqrt{t}$ bychom získali Fermatovu spirálu, pro volbu $r(t) = \frac{a}{\sqrt{t}}$ pak spirálu Lituovou. Pól spirály je bod, ze kterého by se spirála „vykreslovala“, pokud by byl koefficient $a = 0$. V kartézských souřadnicích je pro neposunutou křivku pólem počátek soustavy souřadnic $[0, 0]$. Pól spirály a počátek spirály jsou u Archimédovy spirály totožné.

⁵⁾ Logaritmická spirála se rychleji rozvíjí (roste hodnota koefficientu b) pro klesající velikost tečného úhlu θ .


Fotografie schránek byly umístěny do kartézské soustavy souřadnic a ručně odečteny souřadnice bodů ležících na obloucích tvořících průmět okraje ulity do roviny, přičemž body byly voleny tak, aby jejich úhlová vzdálenost byla ca 5° . Porovnáním s modelem posuzovaných křivek (Geogebra) byly odhadnuty vstupní hodnoty parametrů křivek (tj. souřadnice středů křivek, hodnoty koeficientů apod.) pro následnou optimalizaci.

Při optimalizaci byly minimalizovány funkce, které vypočítaly součet druhých mocnin vzdáleností (tzv. *minimální kvadratická chyba*) mezi pozorovanými daty a matematickými modely křivek. Odhadnuté hodnoty byly použity k vytvoření matematických modelů křivek, kterými byly proloženy datové soubory.



Obr. 5 Kružík drobný (*Gyraulus riparius*). Foto Michal Horsák:

<http://mollusca.sav.sk/malacology/img/gyraulus-riparius/>



Obr. 6 Menetovník rozšířený (*Menetus dilatatus*). Foto Michal Horsák:


<http://mollusca.sav.sk/malacology/img/menetus-dilatatus/>

Funkce, která vyjadřuje součet druhých mocnin vzdáleností mezi pozorovanými $[x_i, y_i]$ a modelovými hodnotami $[u_i, v_i]$ spirál je popsána rovnicí [5, 7]


$$F_{x_0, y_0, a, b} = \sum_{i=1}^n (x_0 + r(t_i) \cos(t_i) - x_i)^2 + (y_0 + r(t_i) \sin(t_i) - y_i)^2, \quad (2)$$

kde $r(t)$ je funkce s parametry a a b definující daný typ spirály a parametr b se uplatní pouze pro logaritmickou spirálu. Hodnoty t_i byly určeny z podmínky [7]

$$\tan(t_i) = \frac{y_i - y_0}{x_i - x_0}. \quad (3)$$



Obr. 7 Terčovník vroubený (*Planorbis planorbis*). Foto Michal Horsák: <http://mollusca.sav.sk/malacology/img/planorbis-planorbis/>



Obr. 8 Lištovka lesklá (*Segmentina nitida*). Foto Michal Horsák: <http://mollusca.sav.sk/malacology/img/segmentina-nitida/>

4. Modelování v SAS

V této části ukážeme zápis programu pro odhad parametrů logaritmické spirály tvořící schránku svinutce běloústého (Anisus leucostoma, obr. 4). Program pro optimalizaci logaritmické spirály byl sestaven s využitím článku [5]. Získaná data byla analyzována softwarem SAS, který používá optimalizační metodu v SAS/IML k nalezení parametrů zkoumané křivky [5, 6, 7] podle hodnot uspořádaných dvojic $\{[x_1, y_1], \dots, [x_n, y_n]\}$ zjištěných bodů. Sestavené programy analyzovaly zjištěná data metodou TRM (trust region method, metoda důvěryhodné oblasti), tj. numerickou metodou optimalizace využívající nelineární regresi.⁶⁾ Jednotlivé části programového kódu jsou doplněny poznámkami vysvětlujícími smysl dané části programu (`/*...*/`).

```
/*Body tvorici casti spiraly omezene dalsim zavitem.*/
/*V obrazku jsou tyto body vyznaceny cervene.*/
/*V prikladu uveden pouze prvni a posledni datovy bod.*/
data S1;
input Segment x y;
Seg = 37 - _N_ + 1;
datalines;
1    0.38    0.31
...
37   5.30   -2.55
;
/*Body tvorici posledni (neomezenou) cast spiraly.*/
/*V obrazku jsou tyto body vyznaceny modre.*/
/*V prikladu uveden pouze prvni a posledni datovy bod.*/
data S2;
input Segment x y;
Seg = 19 - _N_ + 1;
datalines;
1    4.42   -4.43
...
19   8.54   -2.82
;
```

⁶⁾Více o použité metodě TRM např. https://en.wikipedia.org/wiki/Trust_region. Oficiální stránky společnosti: https://www.sas.com/cs_cz/home.html. Software lze pro výzkumné a studijní účely po registraci používat bezplatně (SAS University Edition). Řešení řady problémů např. na: <https://blogs.sas.com/content/iml/>.

```

/*Procedura pri vypoctu zajistuje cteni dat po segmentech.*/
/*Poradi je urcene cislovanim.*/
proc sort data=S1;
by seg;
run;
proc sort data=S2;
by seg;
run;

/*Simuluje model logaritmické spirály.*/
title "Anisus leucostoma";
title2 "Model logaritmické spirály";
proc iml;
pi = constant("pi");
theta = do(0,6*pi, pi/24)';
a = 1; b = -0.3063;
x = a*cos(theta)#exp(b*theta);
y = a*sin(theta)#exp(b*theta);
call scatter(x,y);

/*Procedura vypocte prirustek hodnoty uhlu pro dany bod.*/
start IncrAngle(x, y);
theta = atan2(y,x); dif = dif(theta);
idx = loc(dif^=. & dif<0);
n = nrow(dif); pi = constant("pi");
do while( ncol(idx)>0 );
s = idx[1];
theta[s:n] = 2*pi + theta[s:n]; dif = dif(theta);
idx = loc(dif^=. & dif<0);
end;
return(theta);
finish;
phi = IncrAngle(x,y);

/***** Hlavni vypocet *****/
use S2; read all var {x y}; close;
start ObjF(p) global(x, y);
x0=p[1]; y0=p[2]; a=p[3]; b=p[4];
xc = x-x0; yc = y-y0;
theta = IncrAngle(xc, yc);
u = a*cos(theta)#exp(b*theta);
v = a*sin(theta)#exp(b*theta);

```



```

d2 = (xc-u)##2 + (yc-v)##2;
return( sum(d2) );
finish;
/*Odhad parametru logaritmické spirály.*/
p = {-0.2 0.4 5 0.05}; opt = {0 1};
call nlpnr(rc,result,"ObjF",p,opt);
b= result[4]; tu=atan(1/b);
pi = constant("pi"); tp=(tu*180/pi);
print rc,result[c={"x0" "y0" "a" "b"} label="Log. spirála"];
print tp[label="Tecny uhel"];
/*Vytvori datove soubory "Fit" pro casti log. spirály.*/
x0=result[1];y0= result[2]; a= result[3];b= result[4];
xc = x-x0; yc = y-y0; pi = constant("pi");
r1=a#(exp(-b*2*pi))##2;
theta = do(0,6*pi, pi/24)';
u1 =x0+ (a##2/r1)*cos(theta)#exp(b*theta);
v1 =y0+ (a##2/r1)*sin(theta)#exp(b*theta);
create Fit1 var {"theta" "u1" "v1"};
append;
close;
x0=result[1];y0= result[2]; a= result[3];b= result[4];
xc = x-x0; yc = y-y0; pi = constant("pi");
r2=a#exp(-b*2*pi);
theta = do(0,6*pi, pi/24)';
u2 =x0+ (a##2/r2)*cos(theta)#exp(b*theta);
v2 =y0+ (a##2/r2)*sin(theta)#exp(b*theta);
create Fit2 var {"theta" "u2" "v2"};
append;
close;
x0=result[1];y0= result[2]; a= result[3];b= result[4];
xc = x-x0; yc = y-y0; pi = constant("pi");
r3=a#exp(b*2*pi);
theta = do(0,6*pi, pi/24)';
u3 =x0+ (a##2/r3)*cos(theta)#exp(b*theta);
v3 =y0+ (a##2/r3)*sin(theta)#exp(b*theta);
create Fit3 var {"theta" "u3" "v3"};
append;
close;
use S2; read all var {x y}; close;
u = x; v = y;
create S1c var {"u" "v"};
append;

```

```

close ;
quit ;
/*Slouci pozorovana data a datove soubory "Fit".*/
data Alltwo ;
set S1 S1c Fit1 Fit2 Fit3 ;
run ;
/*Vytvori graf z pozorovanych dat a datovych souboru "Fit".*/
ods graphics / width=500px height=500px ;
title "Shoda dat s modelem log. spiraly" ;
title "Anisus leucostoma" ;
proc sgplot data=Alltwo NOAUTOLEGEND ;
scatter x=x y=y / markerattrs=(color=red) ;
scatter x=u y=v / markerattrs=(color=blue) ;
series x=u1 y=v1 ;
series x=u2 y=v2 ;
series x=u3 y=v3 ;
xaxis grid values=(-12 to 12 by 1) ;
yaxis grid values=(-12 to 12 by 1) ;
run ;

```

5. Výsledky

Modely spirál byly vytvářeny pomocí bodů ležících na okraji posledního (nepřekrytého) závitů spirály (v modelech vyznačeny modrými kroužky). Okraje vnitřních závitů, které se mohou z určité části překrývat, jsou vyznačeny v modelech červenými kroužky. Sestavené modely (obr. 4, 5, 6, 7) ukazují, že schránky okružákových jsou zavínuté ve tvaru logaritmické spirály. V tab. 1 jsou pro model logaritmické spirály zaznamenané hodnoty parametru b a dále vypočtený tečný úhel spirály. Záporné znaménko uvedené u hodnot parametru b vyjadřuje orientaci spirál, které jsou ve všech případech pravotočivé.

Tab. 1 Tečné úhly logaritmických spirál vybraných druhů okružákovitých

	koeficient b	tečný úhel
svinutec běloústý (<i>Anisus leucostoma</i>)	-0,0622	86,4°
kružník drobný (<i>Gyraulus riparius</i>)	-0,1293	82,6°
menetovník rozšířený (<i>Menetus dilatatus</i>)	-0,1385	82,1°
terčovník vroubený (<i>Planorbis planorbis</i>)	-0,0861	85,1°
lištovka lesklá (<i>Segmentina nitida</i>)	-0,0755	85,7°

Závěr

Vytvořené modely ulit okružáků mají ve všech případech tvar logaritmické spirály. Zajímavou otázkou, kterou by mohl zodpovědět detailnější průzkum provedený na větším vzorku, je úvaha, zda je tečný úhel logaritmických spirál druhově specifický a mohl by být použit jako taxonomický znak.

Poděkování

Autor děkuje prof. RNDr. Michalu Horsákovi, Ph.D. za laskavé svolení k použití fotografií okružákovitých.

Literatura

- [1] *Bonney, M., Coaplen, J., Doeff, E.*: What makes a good friend? The mathematics of rock climbing. *SIAM Rev.* 40 (3) (1998), 674–679.
- [2] *Harary, G., Tal, A.*: The natural 3D spirals. *Computer graphics forum* 30 (2) (2011).
- [3] *Holcombe, S. A., Wang, S. C., Grotberg, J. B.*: Modeling female and male rib geometry with logarithmic spirals. *Journal of Biomechanics* 49 (2016), 2995–3003.
- [4] *Shi, P. J., Huang, J. G., Hui, C., Grissino-Mayer, H. D., Tardif, J. C., Zhai, L. H., Wang, F. S., Li, B. L.*: Capturing spiral radial growth of conifers using the superellipse to model tree-ring geometric shape. *Frontiers in Plant Science* 6 (2015), 856.
- [5] *Wicklin, R.*: The spiral of splatter. 2015. [online.] <https://blogs.sas.com/content/iml/2015/06/11/spiral-of-splatter.html>
- [6] *Wicklin, R.*: Fit a circle to data. 2015. [online.] <https://blogs.sas.com/content/iml/2015/06/08/fit-circle.html>
- [7] *Wicklin, R.*: Computing polar angles from coordinate data. 2015. [online.] <https://blogs.sas.com/content/iml/2015/06/10/polar-angle-curve.html>
- [8] *Spíchal, L.*: Modeling the Shape of Tusks of Wild Boars (*Sus scropha* L.). In: M. Hrubý, P. Račková: *Matematika, informační technologie a aplikované vědy* (MITAV 2020). Univerzita Obrany, Brno, 2020, 9 s.